Study of the Second Element of the Painlevé 2 hierarchy using various Darboux coordinates and making the link explicit connections with isospectral coordinates.

Loading the Lax pair obtained for the second element of the Painlevé 2 hierarchy in "Hamiltonian representation of isomonodromic deformations of general rational connections on $\mathfrak{gl}_2(\mathbb{C})$". In our case we have \tau_{\infty,i}=2t_{\infty,i}. The auxiliary matrices thus also get an extra 2 factor because of this change of time normalization.

```
> restart:
  with(LinearAlgebra):
  tau2:=2*t2;
  tau1:=2*t1;

  tdL:=Matrix(2,2,0):
  tdL[1,1]:=-lambda^3+((p1-p2)/(q1-q2)+q1^2+q2*q1+q2^2)*lambda+(-
  p1*q2+p2*q1)/(q1-q2)-q1*q2*(q1+q2);
  tdL[1,2]:=(lambda-q1)*(lambda-q2):
  tdL[2,1]:= (tau2+(2*(p1-p2))/(q1-q2)+2*q1^2+2*q2*q1+2*q2^2)*
  lambda^2+((q1+q2)*tau2+tau1+(2*(p1*q1-p2*q2))/(q1-q2)+(2*(q1+q2))
  *(q1^2+q2^2))*lambda+(q1+q2)*tau1+(1/4)*tau2^2+(q1^2+q1*q2+q2^2)*
  tau2-(p1-p2)^2/(q1-q2)^2-(2*(p1*q2-p2*q1))*(q1+q2)/(q1-q2)+q1^4+
  q2^4-q1^2*q2^2+2*tinfty10;
  tdL[2,2]:=-tdL[1,1]:
  tdL:

  tdA1:=2*Matrix(2, 2, [[-(1/2)*q1-(1/2)*q2-(1/2)*lambda, 1/2], [
  (tau2*(q1-q2)+2*q1^3-2*q2^3+2*p1-2*p2)/(2*q1-2*q2), (1/2)*q1+
  (1/2)*q2+(1/2)*lambda]]);

  tdA2:=2*Matrix(2, 2, [[(-q2^3-q1*q2^2+(lambda^2+q1^2)*q2+q1^3-
  lambda^2*q1+p1-p2)/(4*q1-4*q2), (1/4)*lambda-(1/4)*q1-(1/4)*q2],
  [(2*q1^4+2*lambda*q1^3+q1^2*tau2+(lambda*tau2+2*p1+tau1)*q1-2*
  q2^4-2*lambda*q2^3-q2^2*tau2+(-lambda*tau2-2*p2-tau1)*q2+2*(p1-
  p2)*lambda)/(4*q1-4*q2), (-q1^3-q1^2*q2+(lambda^2+q2^2)*q1+q2^3-
  lambda^2*q2-p1+p2)/(4*q1-4*q2)]]);


  Ham1:=2*unapply( (p1^2-p2^2)/2/(q1-q2)-(1/2)*(q1+q2)*((q1^2+q2^2)
  ^2-q1^2*q2^2)
  -1/2*(q1^2+q1*q2+q2^2)*tau1-1/2*(q1+q2)*(q1^2+q2^2)*tau2
  -1/8*(q1+q2)*tau2^2 -1/2*(q1+q2)*(2*tinfty10-1), q1,q2,p1,p2);

  Ham2:=2*unapply((1/4)*(q1*p2^2-q2*p1^2)/(q1-q2)+1/4*(q1^4+q1^3*
```

```
q2+q1^2*q2^2+q1*q2^3+q2^4)*q2*q1
-(1/4)*1*(p1-p2)/(q1-q2)
+(1/4*(q1+q2))*q2*q1*tau1+(1/4*(q1^2+q1*q2+q2^2))*q2*q1*
tau2+1/16*q2*q1*tau2^2
+(1/4)*q2*q1*(2*tinfty10-1), q1,q2,p1,p2);

Hinfty1:=Ham1(q1,q2,p1,p2):
Hinfty0:=2*Ham2(q1,q2,p1,p2):

omega:=1;

dq1dt1:=diff(Ham1(q1,q2,p1,p2),p1):
dq2dt1:=diff(Ham1(q1,q2,p1,p2),p2):
dp1dt1:=-diff(Ham1(q1,q2,p1,p2),q1):
dp2dt1:=-diff(Ham1(q1,q2,p1,p2),q2):
dq1dt2:=diff(Ham2(q1,q2,p1,p2),p1):
dq2dt2:=diff(Ham2(q1,q2,p1,p2),p2):
dp1dt2:=-diff(Ham2(q1,q2,p1,p2),q1):
dp2dt2:=-diff(Ham2(q1,q2,p1,p2),q2):
```

$$\tau 2 := 2\, t2 \qquad \textbf{(1)}$$

$$\tau 1 := 2\, t1$$

$$tdL_{1,1} := -\lambda^3 + \left(\frac{p1 - p2}{q1 - q2} + q1^2 + q2\, q1 + q2^2\right)\lambda + \frac{-p1\, q2 + p2\, q1}{q1 - q2} - q1\, q2\, (q1 + q2)$$

$$tdL_{2,1} := \left(2\, t2 + \frac{2\,(p1 - p2)}{q1 - q2} + 2\, q1^2 + 2\, q2\, q1 + 2\, q2^2\right)\lambda^2 + \left(2\,(q1 + q2)\, t2 + 2\, t1 + \frac{2\,(p1\, q1 - p2\, q2)}{q1 - q2} + 2\,(q1 + q2)\left(q1^2 + q2^2\right)\right)\lambda + 2\,(q1 + q2)\, t1 + t2^2 + 2\left(q1^2 + q1\, q2 + q2^2\right) t2 - \frac{(p1 - p2)^2}{(q1 - q2)^2} - \frac{2\,(p1\, q2 - p2\, q1)\,(q1 + q2)}{q1 - q2} + q1^4 + q2^4 - q1^2\, q2^2 + 2\, tinfty10$$

$$\begin{bmatrix} -q1 - q2 - \lambda & 1 \\ \frac{2\left(2\, t2\,(q1 - q2) + 2\, q1^3 - 2\, q2^3 + 2\, p1 - 2\, p2\right)}{2\, q1 - 2\, q2} & q1 + q2 + \lambda \end{bmatrix}$$

$$\left[\left[\frac{2\left(-q2^3 - q1\, q2^2 + \left(\lambda^2 + q1^2\right) q2 + q1^3 - \lambda^2\, q1 + p1 - p2\right)}{4\, q1 - 4\, q2}, \frac{1}{2}\lambda - \frac{1}{2}\, q1 - \frac{1}{2}\, q2\right],\right.$$

$$\left[\frac{1}{4\, q1 - 4\, q2}\left(2\left(2\, q1^4 + 2\,\lambda\, q1^3 + 2\, q1^2\, t2 + \left(2\,\lambda\, t2 + 2\, p1 + 2\, t1\right) q1 - 2\, q2^4\right.\right.\right.$$

$$-2\lambda q2^3-2q2^2 t2+(-2\lambda t2-2p2-2t1)q2+2(p1-p2)\lambda)),$$
$$\frac{2(-q1^3-q1^2 q2+(\lambda^2+q2^2)q1+q2^3-\lambda^2 q2-p1+p2)}{4q1-4q2}]]$$

$$Ham1 := 2\left((q1,q2,p1,p2)\to\frac{1}{2}\frac{p1^2-p2^2}{q1-q2}-\frac{1}{2}(q1+q2)((q2^2+q1^2)^2-q1^2 q2^2)-(q1^2+q2\,q1+q2^2)\,t1-(q1+q2)(q2^2+q1^2)\,t2-\frac{1}{2}(q1+q2)\,t2^2-\frac{1}{2}(q1+q2)(2\,tinfty10-1)\right)$$

$$Ham2 := 2\left((q1,q2,p1,p2)\to\frac{1}{4}\frac{-p1^2 q2+p2^2 q1}{q1-q2}+\frac{1}{4}(q1^4+q1^3 q2+q1^2 q2^2+q1\,q2^3+q2^4)\,q2\,q1-\frac{1}{4}\frac{p1-p2}{q1-q2}+\frac{1}{2}(q1+q2)\,q2\,q1\,t1+\frac{1}{2}(q1^2+q2\,q1+q2^2)\,q2\,q1\,t2+\frac{1}{4}q2\,q1\,t2^2+\frac{1}{4}q2\,q1\,(2\,tinfty10-1)\right)$$

$$\omega := 1$$

First change of Darboux coordiantes: From (q,p) to (Q,P). This change is time-independent and symplectic.

We first compute the change of coordinates and the evolutions of the new variables Qinfty0, Qinfty1, Pinfty0, Pinfty1.

```
> solve({p1=omega*Pinfty0*q2-omega*Pinfty1,p2=omega*Pinfty0*q1-
  omega*Pinfty1},{Pinfty0,Pinfty1});
  q1sol:=(-Qinfty1-sqrt(Qinfty1^2-4*omega*Qinfty0))/2/omega;
  q2sol:=(-Qinfty1+sqrt(Qinfty1^2-4*omega*Qinfty0))/2/omega;
  p1sol:=omega*Pinfty0*q2-omega*Pinfty1;
  p2sol:=omega*Pinfty0*q1-omega*Pinfty1;
  simplify(Qinfty1-(-omega*(q1sol+q2sol)));
  simplify(Qinfty0-(omega*q1sol*q2sol));
  Qinfty1sol:=-omega*(q1+q2):
  Qinfty0sol:=omega*q1*q2:
  Pinfty0sol := -(p1-p2)/(q1-q2):
  Pinfty1sol := -(p1*q1-p2*q2)/(q1-q2):
  simplify(p1sol-(omega*Pinfty0*q2-omega*Pinfty1));
  simplify(p2sol-(omega*Pinfty0*q1-omega*Pinfty1));

  dQinfty1dt1inter:=unapply( simplify( diff(Qinfty1sol,q1)*
  dq1dt1+diff(Qinfty1sol,q2)*dq2dt1
  +diff(Qinfty1sol,p1)*dp1dt1+diff(Qinfty1sol,p2)*dp2dt1+diff
  (Qinfty1sol,t1)),p1,p2):
  dQinfty0dt1inter:=unapply( simplify( diff(Qinfty0sol,q1)*dq1dt1+
  diff(Qinfty0sol,q2)*dq2dt1
  +diff(Qinfty0sol,p1)*dp1dt1+diff(Qinfty0sol,p2)*dp2dt1+diff
```

```
(Qinfty0sol,t1)),p1,p2):
dPinfty1dt1inter:=unapply( simplify( diff(Pinfty1sol,q1)*dq1dt1+
diff(Pinfty1sol,q2)*dq2dt1
+diff(Pinfty1sol,p1)*dp1dt1+diff(Pinfty1sol,p2)*dp2dt1+diff
(Pinfty1sol,t1)),p1,p2):
dPinfty0dt1inter:=unapply( simplify( diff(Pinfty0sol,q1)*dq1dt1+
diff(Pinfty0sol,q2)*dq2dt1
+diff(Pinfty0sol,p1)*dp1dt1+diff(Pinfty0sol,p2)*dp2dt1+diff
(Pinfty0sol,t1)),p1,p2):

dQinfty1dt1inter2:=unapply(simplify(dQinfty1dt1inter(p1sol,p2sol)
),q1,q2):
dQinfty0dt1inter2:=unapply(simplify(dQinfty0dt1inter(p1sol,p2sol)
),q1,q2):
dPinfty1dt1inter2:=unapply(simplify(dPinfty1dt1inter(p1sol,p2sol)
),q1,q2):
dPinfty0dt1inter2:=unapply(simplify(dPinfty0dt1inter(p1sol,p2sol)
),q1,q2):

dQinfty1dt1:=simplify(dQinfty1dt1inter2(q1sol,q2sol));
dQinfty0dt1:=simplify(dQinfty0dt1inter2(q1sol,q2sol));
dPinfty1dt1:=simplify(dPinfty1dt1inter2(q1sol,q2sol));
dPinfty0dt1:=simplify(dPinfty0dt1inter2(q1sol,q2sol));



dQinfty1dt2inter:=unapply( simplify( diff(Qinfty1sol,q1)*dq1dt2+
diff(Qinfty1sol,q2)*dq2dt2
+diff(Qinfty1sol,p1)*dp1dt2+diff(Qinfty1sol,p2)*dp2dt2+diff
(Qinfty1sol,t2)),p1,p2):
dQinfty0dt2inter:=unapply( simplify( diff(Qinfty0sol,q1)*dq1dt2+
diff(Qinfty0sol,q2)*dq2dt2
+diff(Qinfty0sol,p1)*dp1dt2+diff(Qinfty0sol,p2)*dp2dt2+diff
(Qinfty0sol,t2)),p1,p2):
dPinfty1dt2inter:=unapply( simplify( diff(Pinfty1sol,q1)*dq1dt2+
diff(Pinfty1sol,q2)*dq2dt2
+diff(Pinfty1sol,p1)*dp1dt2+diff(Pinfty1sol,p2)*dp2dt2+diff
(Pinfty1sol,t2)),p1,p2):
dPinfty0dt2inter:=unapply( simplify( diff(Pinfty0sol,q1)*dq1dt2+
diff(Pinfty0sol,q2)*dq2dt2
+diff(Pinfty0sol,p1)*dp1dt2+diff(Pinfty0sol,p2)*dp2dt2+diff
(Pinfty0sol,t2)),p1,p2):
```

```
dQinfty1dt2inter2:=unapply(simplify(dQinfty1dt2inter(p1sol,p2sol)
),q1,q2):
dQinfty0dt2inter2:=unapply(simplify(dQinfty0dt2inter(p1sol,p2sol)
),q1,q2):
dPinfty1dt2inter2:=unapply(simplify(dPinfty1dt2inter(p1sol,p2sol)
),q1,q2):
dPinfty0dt2inter2:=unapply(simplify(dPinfty0dt2inter(p1sol,p2sol)
),q1,q2):

dQinfty1dt2:=simplify(dQinfty1dt2inter2(q1sol,q2sol));
dQinfty0dt2:=simplify(dQinfty0dt2inter2(q1sol,q2sol));
dPinfty1dt2:=simplify(dPinfty1dt2inter2(q1sol,q2sol));
dPinfty0dt2:=simplify(dPinfty0dt2inter2(q1sol,q2sol));
```

$$\left\{ Pinfty0 = -\frac{p1 - p2}{q1 - q2},\ Pinfty1 = -\frac{p1\, q1 - p2\, q2}{q1 - q2} \right\} \quad \textbf{(2)}$$

$$q1sol := -\frac{1}{2}\, Qinfty1 - \frac{1}{2}\sqrt{Qinfty1^2 - 4\, Qinfty0}$$

$$q2sol := -\frac{1}{2}\, Qinfty1 + \frac{1}{2}\sqrt{Qinfty1^2 - 4\, Qinfty0}$$

$$p1sol := Pinfty0\, q2 - Pinfty1$$

$$p2sol := Pinfty0\, q1 - Pinfty1$$

$$0$$

$$0$$

$$0$$

$$0$$

$$dQinfty1dt1 := 2\, Pinfty0$$

$$dQinfty0dt1 := 2\, Pinfty0\, Qinfty1 + 2\, Pinfty1$$

$$dPinfty1dt1 := -5\, Qinfty1^4 + (-6\, t2 + 12\, Qinfty0)\, Qinfty1^2 + 4\, Qinfty1\, t1 - t2^2 + 4\, Qinfty0\, t2 - Pinfty0^2 - 3\, Qinfty0^2 - 2\, tinfty10 + 1$$

$$dPinfty0dt1 := 4\, Qinfty1^3 + (4\, t2 - 6\, Qinfty0)\, Qinfty1 - 2\, t1$$

$$dQinfty1dt2 := Pinfty0\, Qinfty1 + Pinfty1$$

$$dQinfty0dt2 := Pinfty0\, Qinfty1^2 - Pinfty0\, Qinfty0 + Pinfty1\, Qinfty1 + \frac{1}{2}$$

$$dPinfty1dt2 := 3\, Qinfty0^2\, Qinfty1 + \left(-2\, Qinfty1^3 - 2\, Qinfty1\, t2 + t1\right) Qinfty0 - Pinfty0^2\, Qinfty1 - Pinfty0\, Pinfty1$$

$$dPinfty0dt2 := -\frac{1}{2}\, Qinfty1^4 + \frac{1}{2}\,(-2\, t2 + 6\, Qinfty0)\, Qinfty1^2 + Qinfty1\, t1 - \frac{1}{2}\, t2^2 + 2\, Qinfty0\, t2 + \frac{1}{2}\, Pinfty0^2 - \frac{3}{2}\, Qinfty0^2 - tinfty10 + \frac{1}{2}$$

We compute the associated Hamiltonians. Since the change of coordinates is time-independent and symplectic and since we know that (q,p) are canonical coordinates, we may simply replace the variables in the former Hamiltonians.

```
> Ham1intermediate:=unapply( simplify(Ham1(q1,q2,p1sol,p2sol)),q1,
q2):
Ham1QP:=simplify(Ham1intermediate(q1sol,q2sol)):
```

```
Ham1QPbis:=2*(Pinfty0*Pinfty1*omega^2+(1/2)*Pinfty0^2*Qinfty1*
omega+(1/2)*Qinfty1^5/omega^5+(1/2)*(omega*tau2-4*Qinfty0)*
Qinfty1^3/omega^4 -(1/2)*Qinfty1^2*tau1/omega^2+3*Qinfty0^2*
Qinfty1/2/omega^3-Qinfty0*Qinfty1*tau2/omega^2+(1/2)*Qinfty1*
(tau2^2/4+2*tinfty10-1) /omega+(1/2)*Qinfty0*tau1/omega) ;
simplify(Ham1QP-Ham1QPbis);

Ham2intermediate:=unapply( simplify(Ham2(q1,q2,p1sol,p2sol)),q1,
q2):
Ham2QP:=simplify(Ham2intermediate(q1sol,q2sol)):

Ham2QPbis:=2*( (1/4)*Pinfty1^2*omega^2+(1/2)*omega*Qinfty1*
Pinfty0*Pinfty1-(1/4)*omega*Qinfty0*Pinfty0^2 +(1/4)*Qinfty1^2*
Pinfty0^2+(1/4)*1*Pinfty0*omega+(1/4)*Qinfty0^3/omega^3-3/4*
Qinfty1^2*Qinfty0^2/omega^4-(1/4)*tau2*Qinfty0^2/omega^2+(1/4)*
Qinfty0*Qinfty1^4/omega^5+(1/4)*Qinfty0*tau2*Qinfty1^2/omega^3-
(1/4)*tau1*Qinfty0*Qinfty1/omega^2+Qinfty0*(tau2^2/4+2*
tinfty10-1)/(4*omega));
simplify(series(Ham2QP-Ham2QPbis,Qinfty1));

simplify(dQinfty1dt1-diff(Ham1QP,Pinfty1));
simplify(dQinfty0dt1-diff(Ham1QP,Pinfty0));
simplify(dPinfty1dt1-(-diff(Ham1QP,Qinfty1)));
simplify(dPinfty0dt1-(-diff(Ham1QP,Qinfty0)));

simplify(dQinfty1dt2-diff(Ham2QP,Pinfty1));
simplify(dQinfty0dt2-diff(Ham2QP,Pinfty0));
simplify(dPinfty1dt2-(-diff(Ham2QP,Qinfty1)));
simplify(dPinfty0dt2-(-diff(Ham2QP,Qinfty0)));
```

$$Ham1QPbis := 2\,Pinfty0\,Pinfty1 + Pinfty0^2\,Qinfty1 + Qinfty1^5 + (2\,t2 - 4\,Qinfty0)\,Qinfty1^3 - 2\,t1\,Qinfty1^2 + 3\,Qinfty0^2\,Qinfty1 - 4\,Qinfty0\,Qinfty1\,t2 + Qinfty1\,(t2^2 + 2\,tinfty10 - 1) + 2\,Qinfty0\,t1 \quad \textbf{(3)}$$

$$0$$

$$Ham2QPbis := \frac{1}{2}\,Pinfty1^2 + Pinfty0\,Pinfty1\,Qinfty1 - \frac{1}{2}\,Pinfty0^2\,Qinfty0 + \frac{1}{2}\,Pinfty0^2\,Qinfty1^2 + \frac{1}{2}\,Pinfty0 + \frac{1}{2}\,Qinfty0^3 - \frac{3}{2}\,Qinfty0^2\,Qinfty1^2 - Qinfty0^2\,t2 + \frac{1}{2}\,Qinfty0\,Qinfty1^4 + Qinfty0\,Qinfty1^2\,t2 - Qinfty0\,Qinfty1\,t1 + \frac{1}{2}\,Qinfty0\,(t2^2 + 2\,tinfty10 - 1)$$

$$0$$

$$0$$

$$0$$

$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$

We may also verify the formula for Hinfty0 and Hinfty1 appearing in the computation.

```
> Hinfty1function:=unapply(Hinfty1,p1,p2):
  Hinfty1intermediate:=unapply( simplify(Hinfty1function(p1sol,
  p2sol)),q1,q2):
  Hinfty1QP:=simplify(Hinfty1intermediate(q1sol,q2sol));

  Hinfty0function:=unapply(Hinfty0,p1,p2):
  Hinfty0intermediate:=unapply( simplify(Hinfty0function(p1sol,
  p2sol)),q1,q2):
  Hinfty0QP:=simplify(Hinfty0intermediate(q1sol,q2sol));
```

$$Hinfty1QP := Qinfty1^5 + (2\,t2 - 4\,Qinfty0)\,Qinfty1^3 - 2\,t1\,Qinfty1^2 + \left(Pinfty0^2 + 3\,Qinfty0^2 - 4\,Qinfty0\,t2 + t2^2 + 2\,tinfty10 - 1\right) Qinfty1 + 2\,Qinfty0\,t1 + 2\,Pinfty0\,Pinfty1 \quad \textbf{(4)}$$

$$Hinfty0QP := Qinfty0^3 + \left(-3\,Qinfty1^2 - 2\,t2\right) Qinfty0^2 + \left(Qinfty1^4 + 2\,Qinfty1^2\,t2 - Pinfty0^2 - 2\,Qinfty1\,t1 + t2^2 + 2\,tinfty10 - 1\right) Qinfty0 + Pinfty0^2\,Qinfty1^2 + 2\,Pinfty0\,Pinfty1\,Qinfty1 + Pinfty1^2 + Pinfty0$$

We may also rewrite the Lax matrix in the variables (Q,P).

```
> tdL11function:=unapply( tdL[1,1],p1,p2):
  tdL11intermediate:=unapply( tdL11function(p1sol,p2sol),q1,q2):
  tdL11QP:=simplify(tdL11intermediate(q1sol,q2sol));

  tdL12function:=unapply( tdL[1,2],p1,p2):
  tdL12intermediate:=unapply( tdL12function(p1sol,p2sol),q1,q2):
  tdL12QP:=simplify(tdL12intermediate(q1sol,q2sol));

  tdL21function:=unapply( tdL[2,1],p1,p2):
  tdL21intermediate:=unapply( tdL21function(p1sol,p2sol),q1,q2):
  tdL21QP:=simplify(tdL21intermediate(q1sol,q2sol));
```

$$tdL11QP := -\lambda^3 + \left(Qinfty1^2 - Pinfty0 - Qinfty0\right)\lambda + (-Pinfty0 + Qinfty0)\,Qinfty1 - Pinfty1 \quad \textbf{(5)}$$

$$tdL12QP := Qinfty1\,\lambda + \lambda^2 + Qinfty0$$

$$tdL21QP := Qinfty1^4 - 2\,Qinfty1^3\,\lambda + \left(2\,\lambda^2 + 2\,Pinfty0 - 4\,Qinfty0 + 2\,t2\right) Qinfty1^2 + \left((-2\,t2 + 4\,Qinfty0)\,\lambda - 2\,t1 + 2\,Pinfty1\right) Qinfty1 + (2\,t2 - 2\,Pinfty0 - 2\,Qinfty0)\,\lambda^2 + (2\,t1 - 2\,Pinfty1)\,\lambda + t2^2 - 2\,Qinfty0\,t2 - Pinfty0^2 + Qinfty0^2 + 2\,tinfty10$$

Computation of the spectral invariants and verification of Theorem 5.2 relating Hamiltonians and spectral invariants

```
> assume(q1>q2):
  lambdaplus:=series(simplify(series(Eigenvalues(tdL)[1],lambda=
  infinity)),lambda=infinity):
```

```
CoefflambdaplusLambda3:=simplify(-residue(lambdaplus/lambda^4,
lambda=infinity));
CoefflambdaplusLambda2:=simplify(-residue(lambdaplus/lambda^3,
lambda=infinity));
CoefflambdaplusLambda1:=simplify(-residue(lambdaplus/lambda^2,
lambda=infinity));
CoefflambdaplusLambda0:=simplify(-residue(lambdaplus/lambda^1,
lambda=infinity));
CoefflambdaplusLambdaMinus1:=simplify(-residue
(lambdaplus/lambda^0,lambda=infinity));
CoefflambdaplusLambdaMinus2:=simplify(-residue(lambdaplus/lambda^
(-1),lambda=infinity)):
CoefflambdaplusLambdaMinus3:=simplify(-residue(lambdaplus/lambda^
(-2),lambda=infinity)):

I1:=CoefflambdaplusLambdaMinus2;
I2:=1/2*CoefflambdaplusLambdaMinus3;

I1function:=unapply(I1,p1,p2):
I2function:=unapply(I2,p1,p2):
I1inter:=unapply(I1function(p1sol,p2sol),q1,q2):
I2inter:=unapply(I2function(p1sol,p2sol),q1,q2):
I1QP:=unapply(simplify(I1inter(q1sol,q2sol)),Qinfty0,Qinfty1,
Pinfty0,Pinfty1);
I2QP:=unapply(simplify(I2inter(q1sol,q2sol)),Qinfty0,Qinfty1,
Pinfty0,Pinfty1);

simplify(Ham1QP-2*I1QP(Qinfty0,Qinfty1,Pinfty0,Pinfty1));
-residue(lambda^(-2)*tdL[1,2]*diff(tdL[1,1]/tdL[1,2],lambda),
lambda=infinity);
-Qinfty1sol;

simplify(Ham2QP-2*I2QP(Qinfty0,Qinfty1,Pinfty0,Pinfty1));
simplify(-residue(lambda^(-1)*tdL[1,2]*diff(tdL[1,1]/tdL[1,2],
lambda),lambda=infinity));
simplify(-1/2*Qinfty0sol+Pinfty0sol/2);
```

$$CoefflambdaplusLambda3 := 1 \qquad \textbf{(6)}$$

$$CoefflambdaplusLambda2 := 0$$

$$CoefflambdaplusLambda1 := t2$$

$$CoefflambdaplusLambda0 := t1$$

$$CoefflambdaplusLambdaMinus1 := tinfty10$$

$$I1 := \frac{1}{2\, q1{\sim} - 2\, q2{\sim}} \left(-q1{\sim}^{6} - 2\, q1{\sim}^{4}\, t2 - 2\, q1{\sim}^{3}\, t1 + \left(-t2^{2} - 2\, tinfty10 \right) q1{\sim}^{2} - 2\, t2\, t1\, q1{\sim}\right.$$

$$+ q2\sim^6 + 2\, q2\sim^4\, t2 + 2\, q2\sim^3\, t1 + \left(t2^2 + 2\, tinfty10\right) q2\sim^2 + 2\, t2\, t1\, q2\sim + p1^2 - p2^2\Big)$$

$$I2 := \frac{1}{2}\, \frac{1}{2\, q1\sim - 2\, q2\sim} \Big(q1\sim^6\, q2\sim + 2\, q1\sim^4\, q2\sim\, t2 + 2\, q1\sim^3\, q2\sim\, t1 + \big(t2^2 + 2\, tinfty10\big)\, q2\sim\, q1\sim^2 + \big(-q2\sim^6 - 2\, q2\sim^4\, t2 - 2\, q2\sim^3\, t1 + \big(-t2^2 - 2\, tinfty10\big)\, q2\sim^2 + p2^2 - t1^2 - 2\, tinfty10\, t2\big)\, q1\sim - q2\sim\, \big(p1^2 - t1^2 - 2\, t2\, tinfty10\big)\Big)$$

$$I1QP := (Qinfty0, Qinfty1, Pinfty0, Pinfty1) \to \frac{1}{2}\, Qinfty1^5 + \frac{1}{2}\, (2\, t2 - 4\, Qinfty0)\, Qinfty1^3 - Qinfty1^2\, t1 + \frac{1}{2}\, \big(Pinfty0^2 + 3\, Qinfty0^2 - 4\, Qinfty0\, t2 + t2^2 + 2\, tinfty10\big)\, Qinfty1 - t2\, t1 + Qinfty0\, t1 + Pinfty0\, Pinfty1$$

$$I2QP := (Qinfty0, Qinfty1, Pinfty0, Pinfty1) \to \frac{1}{4}\, Qinfty0^3 + \frac{1}{4}\, \big(-3\, Qinfty1^2 - 2\, t2\big)\, Qinfty0^2 + \frac{1}{4}\, \big(Qinfty1^4 + 2\, Qinfty1^2\, t2 - Pinfty0^2 - 2\, Qinfty1\, t1 + t2^2 + 2\, tinfty10\big)\, Qinfty0 + \frac{1}{4}\, Pinfty0^2\, Qinfty1^2 + \frac{1}{2}\, Pinfty0\, Pinfty1\, Qinfty1 - \frac{1}{4}\, t1^2 - \frac{1}{2}\, tinfty10\, t2 + \frac{1}{4}\, Pinfty1^2$$

$$2\, t1\, t2 - Qinfty1$$
$$q1\sim + q2\sim$$
$$q1\sim + q2\sim$$
$$-\frac{1}{2}\, Qinfty0 + \frac{1}{2}\, Pinfty0 + \frac{1}{2}\, t1^2 + tinfty10\, t2$$
$$\frac{-q1\sim^2\, q2\sim + q1\sim\, q2\sim^2 - p1 + p2}{q1\sim - q2\sim}$$
$$\frac{-q1\sim^2\, q2\sim + q1\sim\, q2\sim^2 - p1 + p2}{2\, q1\sim - 2\, q2\sim}$$

Second change of Darboux coordinates: From (Q,P) to (Q,R).

Definition of the change of coordinates and its inverse. This change is time-independent bu not symplectic so the evolutions are not canonically

```
> Pinfty0sol2:=Qinfty1^2-Qinfty0-Rinfty1;
  Pinfty1sol2 :=-Qinfty1^3+2*Qinfty0*Qinfty1+Qinfty1*Rinfty1-
  Rinfty0;
  simplify(Rinfty1-( -Pinfty0sol2-Qinfty0+Qinfty1^2));
  simplify(Rinfty0-(-Pinfty1sol2-Pinfty0sol2*Qinfty1+Qinfty0*
  Qinfty1));
  Rinfty1sol:=-Pinfty0-Qinfty0+Qinfty1^2;
  Rinfty0sol:=-Pinfty1-Pinfty0*Qinfty1+Qinfty0*Qinfty1;
```

$$Pinfty0sol2 := Qinfty1^2 - Qinfty0 - Rinfty1 \tag{7}$$
$$Pinfty1sol2 := -Qinfty1^3 + 2\, Qinfty0\, Qinfty1 + Qinfty1\, Rinfty1 - Rinfty0$$
$$0$$
$$0$$
$$Rinfty1sol := Qinfty1^2 - Pinfty0 - Qinfty0$$
$$Rinfty0sol := -Pinfty0\, Qinfty1 + Qinfty0\, Qinfty1 - Pinfty1$$

Expression of the Lax matrix in the coordinates (Q,R).

```
> tdL11QPfunction:=unapply( tdL11QP,Pinfty0,Pinfty1):
```

```
tdL11QR:=simplify( tdL11QPfunction(Pinfty0sol2,Pinfty1sol2));
tdL12QPfunction:=unapply( tdL12QP,Pinfty0,Pinfty1):
tdL12QR:=simplify( tdL12QPfunction(Pinfty0sol2,Pinfty1sol2));
tdL21QPfunction:=unapply( tdL21QP,Pinfty0,Pinfty1):
tdL21QR:=simplify( tdL21QPfunction(Pinfty0sol2,Pinfty1sol2));
```

$$tdL11QR := -\lambda^3 + Rinfty1\,\lambda + Rinfty0 \quad \textbf{(8)}$$

$$tdL12QR := Qinfty1\,\lambda + \lambda^2 + Qinfty0$$

$$tdL21QR := (2\,t2 + 2\,Rinfty1)\,Qinfty1^2 + \big((-2\,t2 - 2\,Rinfty1)\,\lambda - 2\,t1 - 2\,Rinfty0\big)\,Qinfty1 + (2\,t2 + 2\,Rinfty1)\,\lambda^2 + (2\,t1 + 2\,Rinfty0)\,\lambda + t2^2 - 2\,Qinfty0\,t2 - 2\,Qinfty0\,Rinfty1 - Rinfty1^2 + 2\,tinfty10$$

Computation of the evolutions of Rinfty0 and Rinfty1.

```
> dRinfty1dt1inter:=unapply( simplify( diff(Rinfty1sol,Qinfty1)
  *dQinfty1dt1+ diff(Rinfty1sol,Qinfty0)*dQinfty0dt1+diff
  (Rinfty1sol,Pinfty1)*dPinfty1dt1+diff(Rinfty1sol,Pinfty0)
  *dPinfty0dt1+diff(Rinfty1sol,t1) ),Pinfty0,Pinfty1):
  dRinfty0dt1inter:=unapply(simplify( diff(Rinfty0sol,Qinfty1)*
  dQinfty1dt1+ diff(Rinfty0sol,Qinfty0)*dQinfty0dt1+diff
  (Rinfty0sol,Pinfty1)*dPinfty1dt1+diff(Rinfty0sol,Pinfty0)*
  dPinfty0dt1+diff(Rinfty0sol,t1) ),Pinfty0,Pinfty1):
  dQinfty1dt1inter:=unapply(dQinfty1dt1 ,Pinfty0,Pinfty1):
  dQinfty0dt1inter:=unapply(dQinfty0dt1 ,Pinfty0,Pinfty1):

  dRinfty1dt1:=simplify(dRinfty1dt1inter(Pinfty0sol2,Pinfty1sol2));
  dRinfty0dt1:=simplify(dRinfty0dt1inter(Pinfty0sol2,Pinfty1sol2));
  dQinfty1QRdt1:=simplify(dQinfty1dt1inter(Pinfty0sol2,Pinfty1sol2)
  );
  dQinfty0QRdt1:=simplify(dQinfty0dt1inter(Pinfty0sol2,Pinfty1sol2)
  );
```

$$dRinfty1dt1 := (-4\,t2 - 4\,Rinfty1)\,Qinfty1 + 2\,t1 + 2\,Rinfty0 \quad \textbf{(9)}$$

$$dRinfty0dt1 := (2\,t2 + 2\,Rinfty1)\,Qinfty1^2 + (-2\,t1 - 2\,Rinfty0)\,Qinfty1 + t2^2 - 4\,Qinfty0\,t2 - 4\,Qinfty0\,Rinfty1 - Rinfty1^2 + 2\,tinfty10 - 1$$

$$dQinfty1QRdt1 := 2\,Qinfty1^2 - 2\,Qinfty0 - 2\,Rinfty1$$

$$dQinfty0QRdt1 := 2\,Qinfty0\,Qinfty1 - 2\,Rinfty0$$

Rewriting of the quantities in terms of the coordinates (Q,R)

```
> tdA1function11:=unapply(tdA1[1,1],p1,p2):
  tdA1function12:=unapply(tdA1[1,2],p1,p2):
  tdA1function21:=unapply(tdA1[2,1],p1,p2):
  tdA1function22:=unapply(tdA1[2,2],p1,p2):
  tdA1QPintermediate11:=unapply(tdA1function11(p1sol,p2sol),q1,q2):
  tdA1QPintermediate12:=unapply(tdA1function12(p1sol,p2sol),q1,q2):
  tdA1QPintermediate21:=unapply(tdA1function21(p1sol,p2sol),q1,q2):
  tdA1QPintermediate22:=unapply(tdA1function22(p1sol,p2sol),q1,q2):
```

```
tdA1QP:=Matrix(2,2,0):
tdA1QP[1,1]:=simplify(tdA1QPintermediate11(q1sol,q2sol)):
tdA1QP[1,2]:=simplify(tdA1QPintermediate12(q1sol,q2sol)):
tdA1QP[2,1]:=simplify(tdA1QPintermediate21(q1sol,q2sol)):
tdA1QP[2,2]:=simplify(tdA1QPintermediate22(q1sol,q2sol)):
tdA1QP;

tdA2function11:=unapply(tdA2[1,1],p1,p2):
tdA2function12:=unapply(tdA2[1,2],p1,p2):
tdA2function21:=unapply(tdA2[2,1],p1,p2):
tdA2function22:=unapply(tdA2[2,2],p1,p2):
tdA2QPintermediate11:=unapply(tdA2function11(p1sol,p2sol),q1,q2):
tdA2QPintermediate12:=unapply(tdA2function12(p1sol,p2sol),q1,q2):
tdA2QPintermediate21:=unapply(tdA2function21(p1sol,p2sol),q1,q2):
tdA2QPintermediate22:=unapply(tdA2function22(p1sol,p2sol),q1,q2):
tdA2QP:=Matrix(2,2,0):
tdA2QP[1,1]:=simplify(tdA2QPintermediate11(q1sol,q2sol)):
tdA2QP[1,2]:=simplify(tdA2QPintermediate12(q1sol,q2sol)):
tdA2QP[2,1]:=simplify(tdA2QPintermediate21(q1sol,q2sol)):
tdA2QP[2,2]:=simplify(tdA2QPintermediate22(q1sol,q2sol)):
tdA2QP;

tdA1QRfunction11:=unapply(tdA1QP[1,1],Pinfty0,Pinfty1):
tdA1QRfunction12:=unapply(tdA1QP[1,2],Pinfty0,Pinfty1):
tdA1QRfunction21:=unapply(tdA1QP[2,1],Pinfty0,Pinfty1):
tdA1QRfunction22:=unapply(tdA1QP[2,2],Pinfty0,Pinfty1):

tdA1QR:=Matrix(2,2,0):
tdA1QR[1,1]:=simplify(tdA1QRfunction11(Pinfty0sol2,Pinfty1sol2)):
tdA1QR[1,2]:=simplify(tdA1QRfunction12(Pinfty0sol2,Pinfty1sol2)):
tdA1QR[2,1]:=simplify(tdA1QRfunction21(Pinfty0sol2,Pinfty1sol2)):
tdA1QR[2,2]:=simplify(tdA1QRfunction22(Pinfty0sol2,Pinfty1sol2)):
tdA1QR;

tdA2QRfunction11:=unapply(tdA2QP[1,1],Pinfty0,Pinfty1):
tdA2QRfunction12:=unapply(tdA2QP[1,2],Pinfty0,Pinfty1):
tdA2QRfunction21:=unapply(tdA2QP[2,1],Pinfty0,Pinfty1):
tdA2QRfunction22:=unapply(tdA2QP[2,2],Pinfty0,Pinfty1):

tdA2QR:=Matrix(2,2,0):
tdA2QR[1,1]:=simplify(tdA2QRfunction11(Pinfty0sol2,Pinfty1sol2)):
tdA2QR[1,2]:=simplify(tdA2QRfunction12(Pinfty0sol2,Pinfty1sol2)):
```

```
tdA2QR[2,1]:=simplify(tdA2QRfunction21(Pinfty0sol2,Pinfty1sol2)):
tdA2QR[2,2]:=simplify(tdA2QRfunction22(Pinfty0sol2,Pinfty1sol2)):
tdA2QR;

tdLQR:=Matrix(2,2,0):
tdLQR[1,1]:=tdL11QR:
tdLQR[1,2]:=tdL12QR:
tdLQR[2,1]:=simplify(tdL21QR):
tdLQR[2,2]:=-tdLQR[1,1]:
tdLQR;
```

$$\begin{bmatrix} Qinfty1-\lambda & 1 \\ 2\,Qinfty1^2-2\,Pinfty0-2\,Qinfty0+2\,t2 & -Qinfty1+\lambda \end{bmatrix} \quad \textbf{(10)}$$

$$\left[\left[\frac{1}{2}\,Qinfty1^2-\frac{1}{2}\,\lambda^2-\frac{1}{2}\,Pinfty0,\ \frac{1}{2}\,\lambda+\frac{1}{2}\,Qinfty1\right],\right.$$
$$\left[-Qinfty1^3+Qinfty1^2\,\lambda+(-t2+2\,Qinfty0)\,Qinfty1+(t2-Pinfty0-Qinfty0)\,\lambda+t1\right.$$
$$\left.\left.-Pinfty1,\ -\frac{1}{2}\,Qinfty1^2+\frac{1}{2}\,\lambda^2+\frac{1}{2}\,Pinfty0\right]\right]$$

$$\begin{bmatrix} Qinfty1-\lambda & 1 \\ 2\,t2+2\,Rinfty1 & -Qinfty1+\lambda \end{bmatrix}$$

$$\left[\left[-\frac{1}{2}\,\lambda^2+\frac{1}{2}\,Qinfty0+\frac{1}{2}\,Rinfty1,\ \frac{1}{2}\,\lambda+\frac{1}{2}\,Qinfty1\right],\right.$$
$$\left[(-t2-Rinfty1)\,Qinfty1+\lambda\,t2+Rinfty1\,\lambda+t1+Rinfty0,\ \frac{1}{2}\,\lambda^2-\frac{1}{2}\,Qinfty0\right.$$
$$\left.\left.-\frac{1}{2}\,Rinfty1\right]\right]$$

$$\left[\left[-\lambda^3+Rinfty1\,\lambda+Rinfty0,\ Qinfty1\,\lambda+\lambda^2+Qinfty0\right],\right.$$
$$\left[(2\,t2+2\,Rinfty1)\,Qinfty1^2+\big((-2\,t2-2\,Rinfty1)\,\lambda-2\,t1-2\,Rinfty0\big)\,Qinfty1+(2\,t2\right.$$
$$+2\,Rinfty1)\,\lambda^2+(2\,t1+2\,Rinfty0)\,\lambda+t2^2-2\,Qinfty0\,t2-2\,Qinfty0\,Rinfty1-Rinfty1^2$$
$$\left.\left.+2\,tinfty10,\ \lambda^3-Rinfty1\,\lambda-Rinfty0\right]\right]$$

```
> lambdaplusQR:=series(simplify(series(Eigenvalues(tdLQR)[1],
  lambda=infinity)),lambda=infinity):
  CoefflambdaplusLambda3QR:=simplify(-residue
  (lambdaplusQR/lambda^4,lambda=infinity));
  CoefflambdaplusLambda2QR:=simplify(-residue
  (lambdaplusQR/lambda^3,lambda=infinity));
  CoefflambdaplusLambda1QR:=simplify(-residue
  (lambdaplusQR/lambda^2,lambda=infinity));
  CoefflambdaplusLambda0QR:=simplify(-residue
  (lambdaplusQR/lambda^1,lambda=infinity));
  CoefflambdaplusLambdaMinus1QR:=simplify(-residue
```

```
(lambdaplusQR/lambda^0,lambda=infinity));
CoefflambdaplusLambdaMinus2QR:=simplify(-residue
(lambdaplusQR/lambda^(-1),lambda=infinity)):
CoefflambdaplusLambdaMinus3QR:=simplify(-residue
(lambdaplusQR/lambda^(-2),lambda=infinity)):

I1QR:=expand(CoefflambdaplusLambdaMinus2QR);
I2QR:=expand(1/2*CoefflambdaplusLambdaMinus3QR);
```

$$CoefflambdaplusLambda3QR := 1$$
$$CoefflambdaplusLambda2QR := 0$$
$$CoefflambdaplusLambda1QR := t2$$
$$CoefflambdaplusLambda0QR := t1$$
$$CoefflambdaplusLambdaMinus1QR := tinfty10 \quad \textbf{(11)}$$

$$I1QR := Qinfty1^3\, Rinfty1 + t2\, Qinfty1^3 - 2\, Qinfty0\, Qinfty1\, Rinfty1 - 2\, Qinfty0\, Qinfty1\, t2 - Qinfty1^2\, Rinfty0 - t1\, Qinfty1^2 - \frac{1}{2}\, Qinfty1\, Rinfty1^2 + \frac{1}{2}\, t2^2\, Qinfty1 + Qinfty0\, Rinfty0 + Qinfty0\, t1 + tinfty10\, Qinfty1 + Rinfty0\, Rinfty1 - t2\, t1$$

$$I2QR := -\frac{1}{2}\, Qinfty0^2\, Rinfty1 - \frac{1}{2}\, Qinfty0^2\, t2 + \frac{1}{2}\, Qinfty0\, Qinfty1^2\, Rinfty1 + \frac{1}{2}\, Qinfty0\, Qinfty1^2\, t2 - \frac{1}{2}\, Qinfty0\, Qinfty1\, Rinfty0 - \frac{1}{2}\, Qinfty0\, Qinfty1\, t1 - \frac{1}{4}\, Qinfty0\, Rinfty1^2 + \frac{1}{4}\, Qinfty0\, t2^2 + \frac{1}{2}\, Qinfty0\, tinfty10 - \frac{1}{4}\, t1^2 - \frac{1}{2}\, tinfty10\, t2 + \frac{1}{4}\, Rinfty0^2$$

Third change of Darboux coordinates: from (Q,R) to (u,v)

Definition of the isospectral coordinates

```
> Qinfty1:=uinfty1;
  Qinfty0:=uinfty0+t2/2;
  Rinfty1:=-t2+vinfty1;
  Rinfty0:=-t1+vinfty0;
  tdLuv:=simplify(tdLQR);
  I1uv:=simplify(I1QR);
  I2uv:=simplify(I2QR);
```

$$Qinfty1 := uinfty1 \quad \textbf{(12)}$$
$$Qinfty0 := uinfty0 + \frac{1}{2}\, t2$$
$$Rinfty1 := -t2 + vinfty1$$
$$Rinfty0 := -t1 + vinfty0$$

$$\left[\left[-\lambda^3 + (-t2 + vinfty1)\,\lambda - t1 + vinfty0,\ uinfty1\,\lambda + \lambda^2 + uinfty0 + \frac{1}{2}\, t2\right], \left[-vinfty1^2 + \left(2\,\lambda^2 - 2\,\lambda\, uinfty1 + 2\, uinfty1^2 + t2 - 2\, uinfty0\right) vinfty1 + 2\, vinfty0\,\lambda - 2\, uinfty1\, vinfty0 + 2\, tinfty10,\ \lambda^3 + (t2 - vinfty1)\,\lambda + t1 - vinfty0\right]\right]$$

$$I1uv := uinfty1^3\, vinfty1 - uinfty1^2\, vinfty0 + \frac{1}{2}\, \left(-4\, uinfty0\, vinfty1 - vinfty1^2\right.$$

$+ 2\, tinfty10) \, uinfty1 + \frac{1}{2} \, (-2\, t1 + 2\, vinfty0) \, vinfty1 - \frac{1}{2} \, vinfty0 \, (t2 - 2\, uinfty0)$

$I2uv := \frac{1}{8} \, (-t2 - 2\, uinfty0) \, vinfty1^2 + \frac{1}{8} \, (t2 + 2\, uinfty0) \, \big(2\, uinfty1^2 + t2$
$- 2\, uinfty0\big) \, vinfty1 + \frac{1}{8} \, (-4\, uinfty1\, vinfty0 + 4\, tinfty10) \, uinfty0 + \frac{1}{8} \, ($
$-2\, uinfty1\, vinfty0 - 2\, tinfty10) \, t2 - \frac{1}{2} \, t1\, vinfty0 + \frac{1}{4} \, vinfty0^2$

```
> tdA1uv:=simplify(tdA1QR);
  tdA2uv:=simplify(tdA2QR);

  dtdA1uvdlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dtdA1uvdlambda[i,j]:=
  diff(tdA1uv[i,j],lambda): od: od:
  dtdA2uvdlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dtdA2uvdlambda[i,j]:=
  diff(tdA2uv[i,j],lambda): od: od:
  dtdA1uvdlambda:
  dtdA2uvdlambda:

  dtdLuvdt1:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dtdLuvdt1[i,j]:=
  simplify(
  diff(tdLuv[i,j],uinfty0)*duinfty0dt1+diff(tdLuv[i,j],uinfty1)*
  duinfty1dt1+diff(tdLuv[i,j],vinfty0)*dvinfty0dt1+diff(tdLuv[i,j],
  vinfty1)*dvinfty1dt1 +diff(tdLuv[i,j],t1)
  ): od: od:

  dtdLuvdt2:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dtdLuvdt2[i,j]:=
  diff(tdLuv[i,j],uinfty0)*duinfty0dt2+diff(tdLuv[i,j],uinfty1)*
  duinfty1dt2+diff(tdLuv[i,j],vinfty0)*dvinfty0dt2+diff(tdLuv[i,j],
  vinfty1)*dvinfty1dt2 +diff(tdLuv[i,j],t2): od: od:

  CompatibilityEquation1:=simplify(dtdLuvdt1-dtdA1uvdlambda+
  Multiply(tdLuv,tdA1uv)-Multiply(tdA1uv,tdLuv)):
  CompatibilityEquation2:=simplify(dtdLuvdt2-dtdA2uvdlambda+
  Multiply(tdLuv,tdA2uv)-Multiply(tdA2uv,tdLuv)):
```

$$\begin{bmatrix} uinfty1 - \lambda & 1 \\ 2\, vinfty1 & -uinfty1 + \lambda \end{bmatrix} \qquad \textbf{(13)}$$

$$\begin{bmatrix} -\frac{1}{2}\lambda^2 + \frac{1}{2}\,uinfty0 - \frac{1}{4}\,t2 + \frac{1}{2}\,vinfty1 & \frac{1}{2}\lambda + \frac{1}{2}\,uinfty1 \\ (-uinfty1 + \lambda)\,vinfty1 + vinfty0 & \frac{1}{2}\lambda^2 - \frac{1}{2}\,uinfty0 + \frac{1}{4}\,t2 - \frac{1}{2}\,vinfty1 \end{bmatrix}$$

Computation of the evolutions of the Darboux coordinates (u,v)

```
> simplify(series(CompatibilityEquation1[1,1],lambda=0)):
simplify(series(CompatibilityEquation1[1,2],lambda=0)):
simplify(series(CompatibilityEquation1[2,1],lambda=0)):
simplify(series(CompatibilityEquation1[2,2]
+CompatibilityEquation1[1,1],lambda=0));
dvinfty1dt1:=-4*uinfty1*vinfty1+2*vinfty0;
dvinfty0dt1:=-vinfty1^2+2*(uinfty1^2-2*uinfty0)*vinfty1-2*
uinfty1*vinfty0+2*tinfty10;
duinfty1dt1:=2*uinfty1^2+t2-2*uinfty0-2*vinfty1;
duinfty0dt1:=(t2+2*uinfty0)*uinfty1+2*t1-2*vinfty0;
simplify(series(CompatibilityEquation1[1,1],lambda=0));
simplify(series(CompatibilityEquation1[1,2],lambda=0));
simplify(series(CompatibilityEquation1[2,1],lambda=0));
simplify(series(CompatibilityEquation1[2,2]+
CompatibilityEquation1[1,1],lambda=0));

simplify(series(CompatibilityEquation2[1,1],lambda=0)):
simplify(series(CompatibilityEquation2[1,2],lambda=0)):
simplify(series(CompatibilityEquation2[2,1],lambda=0)):
simplify(series(CompatibilityEquation2[2,2]+
CompatibilityEquation2[1,1],lambda=0));

dvinfty1dt2:=vinfty1*uinfty1^2-uinfty1*vinfty0-(1/2)*vinfty1^2-2*
uinfty0*vinfty1+tinfty10;
dvinfty0dt2:=uinfty1^3*vinfty1-uinfty1^2*vinfty0-(1/2)*(-2*t2*
vinfty1+vinfty1^2-2*tinfty10)*uinfty1-(1/2)*(t2+2*uinfty0)*
vinfty0;
duinfty1dt2:=(1/2)*(t2+2*uinfty0)*uinfty1+t1-vinfty0;
duinfty0dt2:=(t1-vinfty0)*uinfty1-(1/4)*t2^2+(1/2)*t2*vinfty1+
uinfty0^2+uinfty0*vinfty1;

simplify(series(CompatibilityEquation2[1,1],lambda=0));
simplify(series(CompatibilityEquation2[1,2],lambda=0));
simplify(series(CompatibilityEquation2[2,1],lambda=0));
simplify(series(CompatibilityEquation2[2,2]+
CompatibilityEquation2[1,1],lambda=0));
```

$$0 \qquad (14)$$

$$dvinfty1dt1 := -4\, uinfty1\, vinfty1 + 2\, vinfty0$$
$$dvinfty0dt1 := -vinfty1^2 + 2\left(uinfty1^2 - 2\, uinfty0\right) vinfty1 - 2\, uinfty1\, vinfty0 + 2\, tinfty10$$
$$duinfty1dt1 := 2\, uinfty1^2 + t2 - 2\, uinfty0 - 2\, vinfty1$$
$$duinfty0dt1 := (t2 + 2\, uinfty0)\, uinfty1 + 2\, t1 - 2\, vinfty0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$dvinfty1dt2 := vinfty1\, uinfty1^2 - uinfty1\, vinfty0 - \frac{1}{2}\, vinfty1^2 - 2\, uinfty0\, vinfty1 + tinfty10$$
$$dvinfty0dt2 := uinfty1^3\, vinfty1 - uinfty1^2\, vinfty0 - \frac{1}{2}\left(-2\, t2\, vinfty1 + vinfty1^2 - 2\, tinfty10\right) uinfty1 - \frac{1}{2}\,(t2 + 2\, uinfty0)\, vinfty0$$
$$duinfty1dt2 := \frac{1}{2}\,(t2 + 2\, uinfty0)\, uinfty1 + t1 - vinfty0$$
$$duinfty0dt2 := (t1 - vinfty0)\, uinfty1 - \frac{1}{4}\, t2^2 + \frac{1}{2}\, t2\, vinfty1 + uinfty0^2 + uinfty0\, vinfty1$$
$$0$$
$$0$$
$$0$$
$$0$$

Verification of the isospectral condition \delta_t[\td{L}]= \partial_\lambda \td{A}

```
> diff(tdLuv[1,1],t1)-dtdA1uvdlambda[1,1];
  diff(tdLuv[1,2],t1)-dtdA1uvdlambda[1,2];
  diff(tdLuv[2,1],t1)-dtdA1uvdlambda[2,1];
  diff(tdLuv[2,2],t1)-dtdA1uvdlambda[2,2];

  diff(tdLuv[1,1],t2)-dtdA2uvdlambda[1,1];
  diff(tdLuv[1,2],t2)-dtdA2uvdlambda[1,2];
  diff(tdLuv[2,1],t2)-dtdA2uvdlambda[2,1];
  diff(tdLuv[2,2],t2)-dtdA2uvdlambda[2,2];
```

$$0 \qquad (15)$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$

Final change of coordinates to obtain a canonical set of isospectral coordinates (x_2,x_3,y_2,y_3) inspired by Appendix A of "Hamiltonian structure of rational isomonodromic deformation systems" by Bertola, Harnad and Hurtubise.

Definition of the change od Darboux coordinates.

```
> y2sol:=alpha*vinfty1;
  x3sol:=-beta*uinfty1;
  x2sol:=delta*uinfty0+epsilon*vinfty1;
  y3sol:=mu*vinfty0-nu*uinfty1*vinfty1;
  solve({y2=y2sol,y3=y3sol,x2=x2sol,x3=x3sol},{uinfty0,uinfty1,
  vinfty0,vinfty1});
  uinfty0sol :=(alpha*x2-epsilon*y2)/(alpha*delta);
  uinfty1sol :=-x3/beta;
  vinfty0sol :=(alpha*beta*y3-nu*x3*y2)/(alpha*beta*mu);
  vinfty1sol :=y2/alpha;

  dy2dt1function:=unapply( simplify(diff(y2sol,uinfty0)*
  duinfty0dt1+diff(y2sol,uinfty1)*duinfty1dt1+diff(y2sol,vinfty0)*
  dvinfty0dt1+diff(y2sol,vinfty1)*dvinfty1dt1), uinfty0,uinfty1,
  vinfty0,vinfty1):
  dy3dt1function:=unapply( simplify(diff(y3sol,uinfty0)*
  duinfty0dt1+diff(y3sol,uinfty1)*duinfty1dt1+diff(y3sol,vinfty0)*
  dvinfty0dt1+diff(y3sol,vinfty1)*dvinfty1dt1), uinfty0,uinfty1,
  vinfty0,vinfty1):
  dx2dt1function:=unapply( simplify(diff(x2sol,uinfty0)*
  duinfty0dt1+diff(x2sol,uinfty1)*duinfty1dt1+diff(x2sol,vinfty0)*
  dvinfty0dt1+diff(x2sol,vinfty1)*dvinfty1dt1), uinfty0,uinfty1,
  vinfty0,vinfty1):
  dx3dt1function:=unapply( simplify(diff(x3sol,uinfty0)*
  duinfty0dt1+diff(x3sol,uinfty1)*duinfty1dt1+diff(x3sol,vinfty0)*
  dvinfty0dt1+diff(x3sol,vinfty1)*dvinfty1dt1), uinfty0,uinfty1,
  vinfty0,vinfty1):

  dy2dt1:=simplify(dy2dt1function(uinfty0sol,uinfty1sol,
  vinfty0sol,vinfty1sol)):
  dy3dt1:=simplify(dy3dt1function(uinfty0sol,uinfty1sol,vinfty0sol,
  vinfty1sol)):
  dx2dt1:=simplify(dx2dt1function(uinfty0sol,uinfty1sol,vinfty0sol,
  vinfty1sol)):
  dx3dt1:=simplify(dx3dt1function(uinfty0sol,uinfty1sol,vinfty0sol,
  vinfty1sol)):

  dy2dt2function:=unapply( simplify(diff(y2sol,uinfty0)*
  duinfty0dt2+diff(y2sol,uinfty1)*duinfty1dt2+diff(y2sol,vinfty0)*
  dvinfty0dt2+diff(y2sol,vinfty1)*dvinfty1dt2), uinfty0,uinfty1,
  vinfty0,vinfty1):
  dy3dt2function:=unapply( simplify(diff(y3sol,uinfty0)*
```

```
duinfty0dt2+diff(y3sol,uinfty1)*duinfty1dt2+diff(y3sol,vinfty0)*
dvinfty0dt2+diff(y3sol,vinfty1)*dvinfty1dt2), uinfty0,uinfty1,
vinfty0,vinfty1):
dx2dt2function:=unapply( simplify(diff(x2sol,uinfty0)*
duinfty0dt2+diff(x2sol,uinfty1)*duinfty1dt2+diff(x2sol,vinfty0)*
dvinfty0dt2+diff(x2sol,vinfty1)*dvinfty1dt2), uinfty0,uinfty1,
vinfty0,vinfty1):
dx3dt2function:=unapply( simplify(diff(x3sol,uinfty0)*
duinfty0dt2+diff(x3sol,uinfty1)*duinfty1dt2+diff(x3sol,vinfty0)*
dvinfty0dt2+diff(x3sol,vinfty1)*dvinfty1dt2), uinfty0,uinfty1,
vinfty0,vinfty1):

dy2dt2:=simplify(dy2dt2function(uinfty0sol,uinfty1sol,vinfty0sol,
vinfty1sol)):
dy3dt2:=simplify(dy3dt2function(uinfty0sol,uinfty1sol,vinfty0sol,
vinfty1sol)):
dx2dt2:=simplify(dx2dt2function(uinfty0sol,uinfty1sol,vinfty0sol,
vinfty1sol)):
dx3dt2:=simplify(dx3dt2function(uinfty0sol,uinfty1sol,vinfty0sol,
vinfty1sol)):
```

$$y2sol := \alpha\, vinfty1 \tag{16}$$
$$x3sol := -\beta\, uinfty1$$
$$x2sol := \delta\, uinfty0 + \epsilon\, vinfty1$$
$$y3sol := -\nu\, uinfty1\, vinfty1 + \mu\, vinfty0$$
$$\left\{uinfty0 = \frac{\alpha\, x2 - \epsilon\, y2}{\alpha\, \delta},\ uinfty1 = -\frac{x3}{\beta},\ vinfty0 = \frac{\alpha\, \beta\, y3 - \nu\, x3\, y2}{\alpha\, \beta\, \mu},\ vinfty1 = \frac{y2}{\alpha}\right\}$$
$$uinfty0sol := \frac{\alpha\, x2 - \epsilon\, y2}{\alpha\, \delta}$$
$$uinfty1sol := -\frac{x3}{\beta}$$
$$vinfty0sol := \frac{\alpha\, \beta\, y3 - \nu\, x3\, y2}{\alpha\, \beta\, \mu}$$
$$vinfty1sol := \frac{y2}{\alpha}$$

Expression of the Hamiltonians in the coordinates (x_i,y_i).

```
> mu:=1;
beta:=-1;
epsilon:=0;
delta:=1;
nu:=mu;
alpha:=-beta*mu/delta;
```

```
Ham1xy:=((((-delta*t2-2*x2)*x3+2*delta*t1*beta)*mu-2*y3*beta*
(delta-epsilon))*alpha*y2+3*y2^2*(mu*epsilon+(1/3)*nu*(delta-
epsilon))*x3)/(alpha*beta*mu)-2*y3*alpha*x2/mu-(2*beta*mu*x3*
tinfty10+2*x3^2*y3)/beta-beta*t2*y3:
simplify( dx2dt1-diff(Ham1xy,y2));
simplify(series(simplify( dy2dt1+diff(Ham1xy,x2)),y2));
simplify(series(simplify( dy3dt1+diff(Ham1xy,x3)),y2));
simplify(series(simplify( dx3dt1-diff(Ham1xy,y3)),y2));


Ham2xy:=((-(1/2)*x3^2+(1/4)*t2+(1/2)*x2)*y2+x3*(t1-y3)-(1/4)*
t2^2+x2^2)*y2+(x3*y3-tinfty10)*x2+(1/2)*t2*y3*x3+(1/2)*y3*(2*t1-
y3)+(1/2)*tinfty10*t2:
simplify( dx2dt2-diff(Ham2xy,y2));
simplify(series(simplify( dy2dt2+diff(Ham2xy,x2)),y2));
simplify(series(simplify( dy3dt2+diff(Ham2xy,x3)),y2));
simplify(series(simplify( dx3dt2-diff(Ham2xy,y3)),y2));
```

$$\mu := 1 \quad (17)$$
$$\beta := -1$$
$$\epsilon := 0$$
$$\delta := 1$$
$$\nu := 1$$
$$\alpha := 1$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$
$$0$$

```
> Ham1xy:=simplify(Ham1xy);
x2sol:=x2sol;
x3sol:=x3sol;
y2sol:=y2sol;
y3sol:=y3sol;
Ham2xy:=expand(simplify(Ham2xy));
```

$$Ham1xy := 2\, x3^2\, y3 + \left(-y2^2 + (t2 + 2\, x2)\, y2 - 2\, tinfty10\right) x3 + (2\, t1 - 2\, y3)\, y2 + y3\, (t2 - 2\, x2) \quad (18)$$
$$x2sol := uinfty0$$
$$x3sol := uinfty1$$
$$y2sol := vinfty1$$
$$y3sol := -uinfty1\, vinfty1 + vinfty0$$

$$Ham2xy := -\frac{1}{2}\, x3^2\, y2^2 + \frac{1}{4}\, y2^2\, t2 + \frac{1}{2}\, x2\, y2^2 + y2\, x3\, t1 - y2\, x3\, y3 - \frac{1}{4}\, y2\, t2^2 + y2\, x2^2 + x2\, x3\, y3 - x2\, tinfty10 + \frac{1}{2}\, t2\, y3\, x3 + y3\, t1 - \frac{1}{2}\, y3^2 + \frac{1}{2}\, tinfty10\, t2$$

Expression of the spectral invariants in the coordinates (x,y) and verification that they recover the Hamiltonians.

```
> I1uvfunction:=unapply(I1uv,uinfty0,uinfty1,vinfty0,vinfty1):
  I1xy:=simplify(I1uvfunction(uinfty0sol,uinfty1sol,vinfty0sol,
  vinfty1sol));
  I2uvfunction:=unapply(I2uv,uinfty0,uinfty1,vinfty0,vinfty1):
  I2xy:=simplify(I2uvfunction(uinfty0sol,uinfty1sol,vinfty0sol,
  vinfty1sol));
  simplify(Ham1xy-(-2)*I1xy);
  simplify(Ham2xy-(-2)*I2xy);
```

$$I1xy := -x3^2\, y3 + \frac{1}{2}\left(y2^2 + (-t2 - 2\, x2)\, y2 + 2\, tinfty10\right) x3 + \frac{1}{2}\,(-2\, t1 + 2\, y3)\, y2 - \frac{1}{2}\, y3\,(t2 - 2\, x2) \tag{19}$$

$$I2xy := \frac{1}{8}\left(2\, x3^2 - t2 - 2\, x2\right) y2^2 + \frac{1}{8}\left((-4\, t1 + 4\, y3)\, x3 + t2^2 - 4\, x2^2\right) y2 - \frac{1}{4}\, y3\,(t2 + 2\, x2)\, x3 + \frac{1}{4}\, y3^2 - \frac{1}{2}\, y3\, t1 - \frac{1}{4}\, tinfty10\,(t2 - 2\, x2)$$

$$0$$

$$0$$

Expression of the Lax matrices in the coordinates (x,y)

```
> tdLxy11function:=unapply(tdLuv[1,1],uinfty0,uinfty1,vinfty0,
  vinfty1):
  tdLxy12function:=unapply(tdLuv[1,2],uinfty0,uinfty1,vinfty0,
  vinfty1):
  tdLxy21function:=unapply(tdLuv[2,1],uinfty0,uinfty1,vinfty0,
  vinfty1):
  tdLxy22function:=unapply(tdLuv[2,2],uinfty0,uinfty1,vinfty0,
  vinfty1):
  tdLxy:=Matrix(2,2,0):
  tdLxy[1,1]:=simplify(tdLxy11function(uinfty0sol,uinfty1sol,
  vinfty0sol,vinfty1sol)):
  tdLxy[1,2]:=simplify(tdLxy12function(uinfty0sol,uinfty1sol,
  vinfty0sol,vinfty1sol)):
  tdLxy[2,1]:=simplify(tdLxy21function(uinfty0sol,uinfty1sol,
  vinfty0sol,vinfty1sol)):
  tdLxy[2,2]:=simplify(tdLxy22function(uinfty0sol,uinfty1sol,
  vinfty0sol,vinfty1sol)):
  tdLxy;
```

```
tdA1xy11function:=unapply(tdA1uv[1,1],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA1xy12function:=unapply(tdA1uv[1,2],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA1xy21function:=unapply(tdA1uv[2,1],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA1xy22function:=unapply(tdA1uv[2,2],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA1xy:=Matrix(2,2,0):
tdA1xy[1,1]:=simplify(tdA1xy11function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA1xy[1,2]:=simplify(tdA1xy12function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA1xy[2,1]:=simplify(tdA1xy21function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA1xy[2,2]:=simplify(tdA1xy22function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA1xy;

tdA2xy11function:=unapply(tdA2uv[1,1],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA2xy12function:=unapply(tdA2uv[1,2],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA2xy21function:=unapply(tdA2uv[2,1],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA2xy22function:=unapply(tdA2uv[2,2],uinfty0,uinfty1,vinfty0,
vinfty1):
tdA2xy:=Matrix(2,2,0):
tdA2xy[1,1]:=simplify(tdA2xy11function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA2xy[1,2]:=simplify(tdA2xy12function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA2xy[2,1]:=simplify(tdA2xy21function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA2xy[2,2]:=simplify(tdA2xy22function(uinfty0sol,uinfty1sol,
vinfty0sol,vinfty1sol)):
tdA2xy;
```

$$\left[\left[-\lambda^3+(-t2+y2)\,\lambda-t1+x3\,y2+y3,\ x3\,\lambda+\lambda^2+x2+\frac{1}{2}\,t2\right],\right.$$
$$\left.\left[-y2^2+\left(2\,\lambda^2+t2-2\,x2\right)y2-2\,x3\,y3+2\,\lambda\,y3+2\,tinfty10,\ \lambda^3+(t2-y2)\,\lambda+t1-x3\,y2-y3\right]\right] \quad (20)$$

$$
\left[\begin{matrix} x3-\lambda & 1 \\ 2\,y2 & \lambda - x3 \end{matrix}\right]
$$

$$
\left[\begin{matrix} -\frac{1}{2}\lambda^2 + \frac{1}{2}\,x2 - \frac{1}{4}\,t2 + \frac{1}{2}\,y2 & \frac{1}{2}\lambda + \frac{1}{2}\,x3 \\ \lambda\,y2 + y3 & \frac{1}{2}\lambda^2 - \frac{1}{2}\,x2 + \frac{1}{4}\,t2 - \frac{1}{2}\,y2 \end{matrix}\right]
$$